DECUS
PROGRAM LIBRARY

DECUS NO.	12-49
TITLE	COLD START DF32 DISK FORMATTER FOR PS/8 ON A PDP-12
AUTHOR	Mario DeNobili Submitted by: Stanley Rabinowitz
COMPANY	Digital Equipment Corporation Maynard, Massachusetts
DATE	September 4, 1971
SOURCE LANGUAGE	ASSEMBLY LANGUAGE

COLD START DF32 DISK FORMATTER FOR PS/8 ON A PDP-12

DECUS Program Library Write-up DECUS NO. 12-49

PREFACE

The following problem arises for users who have a PDP-12 (with LINCtape) and a DF32 disk and who wish to use the PS/8 programming system:

They would like to use the disk as the system device since this expands the capabilities of PS/8 and speeds it up considerably; however, they cannot devote the disk to the exclusive use of PS/8 since other programs (notably the LAP6-DIAL-MS monitor system) require the use of the disk. Recreating the PS/8 disk system from scratch is normally very time consuming. This document explains a method for the user to create a PS/8 disk system from scratch as easily as he can bootstrap into a PS/8 LINCtape system.

INITIAL CREATION

For this program to work, a special PS/8 LINCtape must be created. This need by done only once. After it has been created, it can be used to create a PS/8 system with the DF32 disk as system device any time desired, or may be used in the usual manner to create a PS/8 tape system with itself as system device.

To create this special cold start disk system tape, the following procedure must be carried out:

1. Use the PS/8 Configurator to create the system desired which uses the DF32 disk as system device. Directions to do this can be found in the source listing of the Configurator and requires changing assembly-time parameters in the Configurator. When the desired disk system is created, test it to see that it works and is the configuration you wanted. Then, using PS/8 PIP, move the files that you want initially on the disk onto the disk, from the beginning, in the order desired, leaving no gaps. We recommend the following programs as essential:

ABSLDR.SV
PIP.SV
PAL8.SV
EDIT.SV

The only restriction here is that the total length of these files must be less than or equal to 46 (decimal) records long. When deciding which programs he wants initially on the disk, the user should keep in mind how much space he wants left. (For a 32K disk, if the above 4 programs are on the disk, there will be only 34 decimal records left for the user.)* This can be checked, and the length of the files ascertained by listing the disk directory using PS/8 PIP with the /E option.

2. Mount a PS/8 LINCtape system tape (as opposed to data tape) on LINCtape drive Ø. If necessary, use PIP to move the files around on this tape so that the beginning of the tape directory coincides exactly with the disk directory, (perform an extended listing). Turn write enable on for this drive.

3. On LINCtape drive 1 mount a PS/8 tape containing the file CLDF32.SV and run it by typing the command RUN DTA1 CLDF32**. When this program finishes, control will return to the PS/8 Keyboard Monitor and you will have successfully turned the tape on unit Ø into a disk system tape. You should then test it by destroying the contents of the disk and attempting a cold start disk bootstrap as described in section II.

BOOTSTRAPPING OFF THE DISK SYSTEM TAPE

Now that you have created the special disk system tape, you may perform a cold start disk bootstrap with it any time you want as follows:

---

* If the user has more than one DF32, he will have more room left.

** Its default starting address is 42ØØ in PDP-8 mode.

Mount the PS/8 disk system tape onto LINCtape drive Ø.
Set left switches to 7ØØ. Set right switches to 1Ø.
Hit I/O preset with mode switch set to LINC Mode.
Tape Ø should move. It should stop with AC=7777.
Hit start 2Ø.
Tape Ø should move about 6 times.
PS/8 (disk) keyboard monitor will respond with a dot.

If during this bootstrap operation, the teletype bell rings, and the computer halts, this means there was an I/O error on the disk. The error lights are in the AC. Hit continue to try again. Make sure the disk is not write-locked.

This procedure destroys the previous contents of the disk.

If the right switches were set to Ø instead of 1Ø, the normal tape bootstrap will occur.

WARM START

Once the disk system has been put up, as long as the disk does not get destroyed, you can perform a warm-start disk bootstrap as described in the PS/8 User's Guide. We summarize below:

If core has not been destroyed (last page of core in fields Ø and 1):

To save contents of core: branch to location 76ØØ (field Ø)
Without saving core: branch to location 76Ø5 (field Ø)

If the resident bootstrap has been destroyed in core, key the following into field Ø:

Location	Contents
775Ø	76ØØ
7751	66Ø3
7752	6622
7753	5352
7754	5752

Set left switches to Ø775Ø. Hit I/O preset with mode switch set to PDP-8 mode, even if you're in 8-mode. Hit start-left-switches.

DISTRIBUTION

LINCtape:

The source is in a PS/8 ASCII file called CLDF32.PA
A PAL8 listing is in a file called CLDF32.LS
The binary is in a PS/8 binary file called CLDF32.BN
A core image file is called CLDF32.SV

EPILOGUE

It is hoped that if the user has a PDP-12 and a different disk, such as an RFØ8, this document will help him implement a similar bootstrapping procedure for his disk.

```
/  COLD START DF32 FORMATTER
/
/       P?S-12-9.1A-UA
/
/       MARIO DENOBILI © P?S
/
/DISK BOOTSTRAP (RECORD 0) MUST RESIDE ON TAPE IN RECORD 5.
/(400 WORDS LONG)
/DISK DIRECTORY (STARTING AT RECORD 1) MUST RESIDE ON TAPE
/IN PAGE 11 (SECOND HALF OF RECORD 5) (200 WORDS LONG)
/THE FIRST PAGE OF THIS PROGRAM MUST RESIDE ON TAPE PAGE 10 (OCTAL)
/(FIRST HALF OF RECORD 4) (200 WORDS LONG)
/
/RESTRICTION:
/TAPE CATALOG MUST NOT EXCEED 3 RECORDS IN LENGTH
/(APPROXIMATELY 150 DECIMAL ENTRIES) AND
/MUST START WITH ABSLDR.SV, PIP.SV, EDIT.SV, PAL8.SV
/IN THE ORDER SPECIFIED ON DISK DIRECTORY.
/
/COLD START FORMATTER PUTS DISK DIRECTORY AND BOOTSTRAP
/ON DISK AND THEN COPIES FIRST 140 OCTAL RECORDS FROM TAPE
/BEGINNING AT RECORD 6 ONTO DISK.
/
/COLD START DF32 FORMATTING PROCEDURE:
/
/       MOUNT TAPE ON LINCTAPE DRIVE 0.
/       HIT I/O PRESET WHILE IN LINC MODE.
/       SET LS=700, RS=10.
/       PRESS DO.  TAPE SHOULD MOVE.
/       WHEN TAPE STOPS, AC SHOULD = 7777.
/       PRESS START 20.
/
/       IF TELETYPE BELL RINGS, THIS MEANS THERE WAS
/       AN ERROR WHILE TRYING TO WRITE ON THE DISK;
/       HIT CONTINUE TO RETRY.
/
/       IF BOOTSTRAP IS SUCCESSFUL,
/       THE KEYBOARD MONITOR SHOULD RESPOND WITH A DOT.
/
/       IF, IN THE ABOVE PROCEDURE, THE RIGHT SWITCHES
/       ARE SET TO 0 INSTEAD OF 10, THE NORMAL TAPE
/       BOOTSTRAP WILL OCCUR.
/
```

```
                /
                /       PDP-12 MNEMONICS
                /
                        PDP=2
                        CLR=11
                        AXO=1
                        TMA=23
                        MTB=703
                        RDC=700
                        WCG=705
                        LINC=6141
                /
                        *4000
4000    0014    L14,    14
4001    0000    TEMP,   0
4002    3700    L3700,  3700
4003    0200    L200,   200
4004    0040    L40,    40
4005    7772    M6,     -6
4006    0000    KNT,    0
4007    7750    L7750,  7750
4010    7751    L7751,  7751
4011    0000    CORE,   0
4012    0000    RKNT,   0
4013    0000    SAVE,   0
4014    0000    CORE2,  0
4015    0207    BELL,   207
4016    0010    L10,    10
```

```
                        *4020           /  I/O PRESET == START 20
4020  0002              PDP
4021  6141              LINC
4022  1020              1020            /LDA I
4023  1020              1020            /AXO OPTIONS
4024  0001              AXO
4025  0002              PDP
4026  7200              CLA
4027  4271              JMS READ        /READ DISK DIRECTORY
4030  0400              400             /INTO CORE PAGE 2
4031  7777              -1              /FROM TAPE PAGE 11
4032  0011              11
4033  4271              JMS READ        /READ DISK BOOTSTRAP
4034  0000              0               /INTO CORE PAGES 0 & 1
4035  7776              -2              /FROM TAPE
4036  0012              12              /PAGES 12 & 13
4037  3250              DCA OUTBLK
4040  4333              JMS DISK        /WRITE TAPE PAGES 0-37
4041  1200              TAD L14
4042  3250              DCA OUTBLK
4043  1205              TAD M6
4044  3206              DCA KNT
4045  4271  LOOP,       JMS READ        /READ 40 PAGES FROM TAPE
4046  0000              0               /INTO FIELD 1
4047  7740              -40
4050  0000  OUTBLK,     0
4051  4333              JMS DISK
4052  1250              TAD OUTBLK      /INCREMENT BLOCK #
4053  1204              TAD L40
4054  3250              DCA OUTBLK
4055  2206              ISZ KNT
4056  5245              JMP LOOP
4057  7200  BOOT,       CLA
4060  6615              DEAL            /RAW DEAL, SIMULATE START
4061  1361              TAD L7600
4062  3607              DCA I L7750     /SET WORD COUNT
4063  1265              TAD L6603
4064  3610              DCA I L7751     /SET CURRENT ADDRESS
4065  6603  L6603,      DMAR            /READ DISK PAGE 0 INTO CORE
4066  6622  L6622,      DFSC            /BEGINNING AT ABSOLUTE CORE LOC 6604
4067  5266              JMP .-1         /I.E. EXECUTE DF32 BOOTSTRAP
4070  5666              JMP I L6622
```

```
                /**********)**********)*******/
                /                              /
                /SUBROUTINE TO READ PDP-12 LINCTAPE
                /FORMATTED WITH 200 OR 201 WORDS PER BLOCK
                /                              /
                /CALLING SEQUENCE:             /
                /                              /
                /       JMS READ               /
                /       CORE LOC. OF INPUT BUFFER START
                /       NEGATIVE OF # OF PAGES TO READ
                /       STARTING BLOCK NUMBER OF TAPE
                /                              /
                /**********)*******************/
                /
4071  0000      READ,   0
4072  1671              TAD I READ
4073  3211              DCA CORE
4074  2271              ISZ READ
4075  1671              TAD I READ
4076  3212              DCA RKNT
4077  2271              ISZ READ
4100  1671              TAD I READ
4101  3315              DCA IBLKNO
4102  6211              CDF 10
4103  2271              ISZ READ
4104  1211      RLOOP,  TAD CORE
4105  1203              TAD L200
4106  3214              DCA CORE2
4107  1614              TAD I CORE2
4110  3213              DCA SAVE
4111  1211              TAD CORE
4112  6141              LINC
4113  0023              TMA
4114  0700              RDC
4115  0000      IBLKNO, 0                       /TAPE BLOCK #
4116  0002              PDP
4117  7001              IAC
4120  7640              SZA CLA
4121  7402              HLT                     /HARDWARE FAILURE
4122  1214              TAD CORE2
4123  3211              DCA CORE
4124  2315              ISZ IBLKNO
4125  2212              ISZ RKNT
4126  5304              JMP RLOOP
4127  1213              TAD SAVE
4130  3614              DCA I CORE2
4131  6201              CDF 0
4132  5671              JMP I READ
```

```
                /********************)********************)*******/
                /                                                  /
                /THIS ROUTINE WRITES OUT ALL OF FIELD 1            /
                /ONTO THE DISK BEGINNING AT THE DISK PAGE          /
                /SPECIFIED BY OUTBLK.                              /
                /                                                  /
                /IF AN ERROR OCCURS, 3 RETRIES ARE MADE.           /
                /IF IT STILL PERSISTS, PROGRAM RINGS THE TTY       /
                /BELL AND HALTS WITH DISK ERROR LIGHTS IN AC.      /
                /HITTING CONTINUE WILL TRY AGAIN 3 MORE TIMES.     /
                /                                                  /
                /********************)*****************************/
                /
4133  0000      DISK,   0
4134  7346              CLA CLL CMA RTL /RETRY 3 TIMES
4135  3201              DCA TEMP
4136  7240      RETRY,  CLA CMA
4137  3610              DCA I L7751     /SET CURRENT ADDRESS
4140  3607              DCA I L7750     /SET WORD COUNT
4141  1250              TAD OUTBLK
4142  7104              CLL RAL
4143  0202              AND L3700
4144  1216              TAD L10
4145  6615              DEAL
4146  7200              CLA
4147  1250              TAD OUTBLK
4150  7112              CLL RTR
4151  7012              RTR
4152  7012              RTR
4153  0361              AND L7600
4154  6605      DFUN,   DMAW
4155  6621              DFSE
4156  5363              JMP ERROR
4157  6622              DFSC
4160  5355              JMP .-3
4161  7600      L7600,  7600            /CLA
4162  5733              JMP I DISK
4163  2201      ERROR,  ISZ TEMP
4164  5336              JMP RETRY
4165  1215              TAD BELL
4166  6046              TLS
4167  7200              CLA
4170  6616              DEAC
4171  7402              HLT
4172  5334              JMP DISK+1
```

```
                        *4200
4200  7300  CREATE,  CLA  CLL
4201  1247           TAD      DDMAR
4202  3656           DCA I    PDFUN
4203  3660           DCA I    POUTBLK
4204  4657           JMS I    PDISK
4205  1250           TAD      DDMAW
4206  3656           DCA I    PDFUN
4207  1253           TAD      L4377
4210  3011           DCA      11
4211  1252           TAD      L377
4212  4231           JMS      MOVE
4213  7240           CLA  CMA
4214  4231           JMS      MOVE
4215  1010           TAD      10
4216  4231           JMS      MOVE
4217  6141  WCGERR,  LINC
4220  0001           AXO
4221  0705           WCG
4222  3010           3010
4223  0002           PDP
4224  7040           CMA
4225  7640           SZA  CLA
4226  5217           JMP      WCGERR
4227  5630           JMP I    .+1
4230  4057           BOOT
                     /
4231  0000  MOVE,    0
4232  3010           DCA      10
4233  1255           TAD      M200
4234  3254           DCA      MOVCNT
4235  6211  MOVLUP,  CDF      10
4236  1410           TAD I    10
4237  6201           CDF      00
4240  3411           DCA I    11
4241  2254           ISZ      MOVCNT
4242  5235           JMP      MOVLUP
4243  1011           TAD      11
4244  1251           TAD      L0200
4245  3011           DCA      11
4246  5631           JMP I    MOVE
                     /
4247  6603  DDMAR,   DMAR
4250  6605  DDMAW,   DMAW
4251  0200  L0200,   200
4252  0377  L377,    377
4253  4377  L4377,   4377
4254  0000  MOVCNT,  0
4255  7600  M200,    -200
4256  4154  PDFUN,   DFUN
4257  4133  PDISK,   DISK
4260  4050  POUTBLK, OUTBLK
                     $
```

AXO	0001	BELL	4015	BOOT	4057	CLR	0011	CORE	4011
CORE2	4014	CREATE	4200	DDMAR	4247	DDMAW	4250	DFUN	4154
DISK	4133	ERROR	4163	IBLKNO	4115	KNT	4006	LINC	6141
LOOP	4045	L0200	4251	L10	4016	L14	4000	L200	4003
L3700	4002	L377	4252	L40	4004	L4377	4253	L6603	4065
L6622	4066	L7600	4161	L7750	4007	L7751	4010	MOVCNT	4254
MOVE	4231	MOVLUP	4235	MTB	0703	M200	4255	M6	4005
OUTBLK	4050	PDFUN	4256	PDISK	4257	PDP	0002	POUTBL	4260
RDC	0700	READ	4071	RETRY	4136	RKNT	4012	RLOOP	4104
SAVE	4013	TEMP	4001	TMA	0023	WCG	0705	WCGERR	4217